
1   ; 
2   ; Z80 dallamcsengő eredeti vezérlőprogramjának visszafejtése.
3   ; (Ez nincs kifejtve a cikkben.)
4   ; 
5   ; Publikálta: Kinczel András, a Rádiótechnika 1987/2 számában. 
6   ;   Dallamcsengő mikroprocesszorral (99.-101. oldalakon)
7   ; 
8   ; Szerző: TCsabI©  
9   ;  Dátum: 2017. július 27.

10   ; E-mail: toth.csaba.hip@gmail.com
11   ;
12   0000 ED56 IM 1 ; A CPU Interrupt Mode 1-be állítása.
13   ;  Az INT bemenétre egy RST 38H
14   ;  (azaz CALL 0038H) utasítást fog végrehajtani.
15   ;  Az NMI bemenétre egy RST 66H
16   ;  (azaz CALL 0066H) utasítást fog végrehajtani.
17   ; Az INT és NMI bemeneteken vannak,
18   ; a dallamindító gombok.
19   ; Visszatérési pont, a dallamok ismételt lejátszásához.
20   0002 219F00 LD HL,009FH ; HL = 009FH
21   ;   A dallamok kezdőcímének inicializálása.
22   ; (Azért nem 00A0h, mert a HALT után van egy
23   ;  INC HL utasítás.)
24   ; Visszatérési pont, a következő dallam lejátszásához.
25   0005 FB EI ; Megszakítások engedélyezése.
26   0006 AF XOR A ; A = A xor A, azaz Akkumulátor nullázása 00h.
27   0007 76 HALT ; CPU működésének felfüggesztése,
28   ; a következő megszakítás kérésig.
29   ; Azaz várakozásás, valamelyik gomb megnyomására.
30   ; Visszatérési pont, a következő hang lejátszásához.
31   0008 23 INC HL ; HL = HL + 1
32   0009 47 LD B,A ; B = A
33   000a 57 LD D,A ; D = A
34   000b 7E LD A,(HL) ; A = (HL), A dallam hangjának kiolvasása
35   ;           a memóriából.
36   000c 5F LD E,A ; E = A, Dallam hangjának mentése E-be.
37   000d E61F AND 1FH ; A = A and 1Fh 00011111 Frekvencia kimaszkolása.
38   000f F680 OR 80H ; A = A or 80h  10000000 Így lesz frekvencia cím.
39   0011 4F LD C,A ; C = A, Így BC regiszterben van a frekvencia.
40   0012 7B LD A,E ; A = E, A dallam hangjának kiolvasása E-ből.
41   0013 07 RLCA ; A eltolása balra 1 bittel.
42   0014 07 RLCA ; A 7.bit -> 0.bitbe töltődik.
43   0015 07 RLCA ; 
44   0016 E607 AND 07H ; A = A and 07h 00000111 Hanghossz kimaszkolása.
45   0018 F630 OR 30H ; A = A or 30h  00110000 Így lesz hanghossz cím. 
46   001a 5F LD E,A ; E = A, Így a DE regiszterben van a hang hossza.
47   001b 1A LD A,(DE) ; A = (DE), A hang hosszának kiolvasása, 
48   ;           a táblázatból.  
49   001c 57 LD D,A ; D = A
50   001d 0A LD A,(BC) ; A = (BC), A frekvencia kiolvasása,
51   ;           a táblázatból.
52   001e FE00 CP 00H ; A == 00h?, azaz A-00h --> Z bit beállítása.
53   0020 2819 JR Z,003BH ; Relatív ugrás a 003Bh címre, ha Z=1. Szünet...
54   0022 FEFE CP 0FEH ; A == FEh?, azaz A-FEh --> Z bit beállítása.
55   0024 28DF JR Z,0005H ; Relatív ugrás a 0005h címre, ha Z=1.
56   ; Következő dallam lejátszása.
57   0026 FEFF CP 0FFH ; A == FFh?, azaz A-FFh --> Z bit beállítása.
58   0028 28D8 JR Z,0002H ; Relatív ugrás a 0002h címre, ha Z=1.
59   ; Dallamkészlet ismételt lejátszása.
60   002a 4F LD C,A ; C = A
61   002b AF XOR A ; A = A xor A, azaz Akkumulátor nullázása 00h.
62   002c 5F LD E,A ; E = A
63   002d C34600 JP 0046H ; Ugrás a 0046h címre. Hang generálása...
64   ;--------------------------------------
65   ; A hangok hosszúságának táblázata.
66   ;--------------------------------------
67   0030 FF ; Egész
68   0031 C1 ; 3/4
69   0032 81 ; 1/2
70   0033 61 ; 3/8
71   0034 41 ; 1/4
72   0035 31 ; 3/16
73   0036 21 ; 1/8



74   0037 11 ; 1/16
75   ;--------------------------------------
76   ; A CPU INT bemenetére megszakításkérés érkezett.
77   ; A dallamcsengő INT bementén lévő gombját megnyomták. Dallam lejátszása.
78   0038 C30800 JP 0008H ; Ugrás a 0008h címre. Következő hang lejátszása.
79   ; Hang nélküli várakozás. Azaz szünet lejátszása... DE időtartammal.
80   003b 0601 LD B,01H ; B = 01h
81   ;*
82   003d 10FE DJNZ 003DH ; B = B - 1, ha B nem lett 0,
83   ;          akkor relatív ugrás a 003Dh címre.
84   003f 1B DEC DE ; DE = DE - 1
85   0040 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
86   0041 20F8 JR NZ,003BH ; Relatív ugrás a 003Bh címre, ha Z=0.
87   0043 C30800 JP 0008H ; Ugrás a 0008h címre. Következő hang lejátszása.
88   ; Hang generálása... BC frekvenciával és DE időtartammal.
89   0046 41 LD B,C ; B = C
90   ;
91   0047 1B DEC DE ; DE = DE - 1
92   0048 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
93   0049 28BD JR Z,0008H ; Relatív ugrás a 0008h címre, ha Z=1.
94   ; Következő hang lejátszása.
95   004b 10FA DJNZ 0047H ; B = B - 1, ha B nem lett 0,
96   ;  akkor relatív ugrás a 0047h címre.
97   004d D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
98   ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
99   ; A CPU ekkor, a IORQ kimenetére egy tűimpulzust

100   ; küld. Az adatbuszon érvényes cím van.
101   ; A tűimpulzus, az SN7474-es D tárolóban szim-
102   ; metrikus négyszögjellé alakul. Ezt halljuk.
103   004f C34600 JP 0046H ; Ugrás a 0046h címre.
104   ; A Bim-Bam második hangjának lejátszása...
105   0052 1680 LD D,80H ; D = 80h
106   ;*
107   0054 0624 LD B,24H ; B = 24h
108   ;*
109   0056 1B DEC DE ; DE = DE - 1
110   0057 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
111   0058 2806 JR Z,0060H ; Relatív ugrás a 0060h címre, ha Z=1.
112   ; Szünet lejátszása.
113   005a 10FA DJNZ 0056H ; B = B - 1, ha B nem lett 0,
114   ;            akkor relatív ugrás a 0056h címre.
115   005c D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
116   ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
117   ; A CPU ekkor, a IORQ kimenetére egy tűimpulzust
118   ; küld. Az adatbuszon érvényes cím van.
119   ; A tűimpulzus, az SN7474-es D tárolóban szim-
120   ; metrikus négyszögjellé alakul. Ezt halljuk.
121   005e 18F4 JR 0054H ; Relatív ugrás a 0054h címre.
122   ; Várakozás...
123   0060 06FF LD B,0FFH ; B = FFh
124   ;*
125   0062 10FE DJNZ 0062H ; B = B - 1, ha B nem lett 0,
126   ;            akkor relatív ugrás a 0062h címre.
127   0064 189F JR 0005H ; Relatív ugrás a 0005h címre.
128   ; Visszaugrás a dallam lejátszási üzemmódba.
129   ;--------------------------------------
130   ; A CPU NMI bemenetére megszakításkérés érkezett.
131   ; A dallamcsengő NMI bementén lévő gombját megnyomták.
132   ; A Bim-Bam első hangjának lejátszása...
133   0066 1620 LD D,20H ; D = 20h
134   ; 
135   0068 0620 LD B,20H ; B = 20h
136   ;*
137   006a 1B DEC DE ; DE = DE - 1
138   006b BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
139   006c 2806 JR Z,0074H ; Relatív ugrás a 0074h címre, ha Z=1.
140   ; Szünet lejátszása. 
141   006e 10FA DJNZ 006AH ; B = B - 1, ha B nem lett 0,
142   ;            akkor relatív ugrás a 0062h címre.
143   0070 D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
144   ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
145   ; A CPU ekkor, a IORQ kimenetére egy tűimpulzust
146   ; küld. Az adatbuszon érvényes cím van.



147   ; A tűimpulzus, az SN7474-es D tárolóban szim-
148   ; metrikus négyszögjellé alakul. Ezt halljuk.
149   0072 18F4 JR 0068H ; Relatív ugrás a 0068h címre.
150   ; Várakozás...
151   0074 06FF LD B,0FFH ; B = FFh
152   ;*
153   0076 10FE DJNZ 0076H ; B = B - 1, ha B nem lett 0,
154   ; akkor relatív ugrás a 0076h címre.
155   0078 C35200 JP 0052H ; Ugrás a 0052h címre.
156   ;--------------------------------------
157   007b FF ;
158   007c FF ;
159   007d FF ; Nem használt terület.
160   007e FF ;
161   007f FF ;
162   ;--------------------------------------
163   ; A hangok frekvenciájának táblázat...
164   ; A frekvencia és a várakozások időtartama, a CPU órajelétől függ! 
165   ;-----------------------------------------------------------------------------------
166   0080 00 ; Nincs hang, azaz szünet.
167   ;--------------------------------------
168   0081 11 ; H' +
169   0082 12 ; Aisz' + 
170   0083 13 ; A' +
171   0084 14 ; Gisz' +
172   0085 15 ; G' +
173   0086 16 ; Fisz' +
174   0087 18 ; F' +
175   0088 19 ; E' +
176   0089 1B ; Disz' +
177   008a 1D ; D' +
178   008b 1E ; Cisz' +
179   008c 20 ; C' +
180   ;--------------------------------------
181   008d 22 ; H +
182   008e 24 ; Aisz előző hang + 2^(1/12) (1 oktáv 12 hang) 
183   008f 26 ; A (normál A hang)   440Hz
184   0090 29 ; Gisz előző hang - 2^(1/12) (1 oktáv 12 hang)
185   0091 2B ; G -
186   0092 2E ; Fisz -
187   0093 31 ; E -
188   0094 34 ; Disz -
189   0095 36 ; D -
190   0095 3A ; Cisz -
191   0097 3D ; C -
192   ;--------------------------------------
193   0098 41 ; 'H -
194   0099 45 ; 'Aisz -
195   009a 4A ; 'A -
196   009b 4D ; 'Gisz -
197   009c 51 ; 'G -
198   009d 56 ; 'Fisz -
199   ;--------------------------------------
200   009e FE ; Dallam vége jelző.
201   009f FF ; Dallamkészlet vége jelző.
202   ;-----------------------------------------------------------------------------------
203   ; Innen kezdődnek a dallamok...
204   00A0 .. .. .. .. .. .. .. .. .. .. .. ..
205   ; A tárolt hang szerkezete...
206   ; Egy bájt, 1 hang.
207   ;       A hang hossza: 7-5 bitek, azaz 3 biten.
208   ; A hang frekvenciája: 4-0 bitek, azaz 5 biten.


